Appendix D 13: Finite element formulation of plane stress plasticity

In [1], we state the basic FEM procedures to deal with the elastic-plastic problem applicable to the general three-dimensional state of stress, plane strain, and an axisymmetric body. These are problems in which all three components of the normal stresses are free variables. In the plane stress, however, the component perpendicular to the midplane of the body is equal to zero, but the component of the stress deviator in this direction is not zero, and these relationships can not be used for this case without some modifications.

1. Plane stress linear elasticity

The plane stress assumption is introduced in the analysis of bodies at which one of the dimensions is much more smaller than the others. Consider such a loaded body in equilibrium (Fig. 1) where the plane (x, y) is the plane of symmetry (midplane *S*) of the body. Let all external loads (including reactions) are distributed symmetrically to the relatively small thickness *t* (*x*, *y*) of the body. Thus the load resultants acting in the midplane: the continuous line load **p**(*x*,*y*) and the body load **b**(*x*,*y*), are independent of the variable *z*. Under these conditions, it is enough to consider only the displacement functions of the points on the midplane of the body

$$\mathbf{u} = \begin{bmatrix} u(x,y) \\ v(x,y) \end{bmatrix} \tag{1}$$



Fig. 1 *The midplane of the body and a 2D finite element (hrúbka = thickness)*

The non-zero components of the stress tensor placed in the column vector $\boldsymbol{\sigma}$ are only the components that lie in the plane (*x*, *y*)

$$\boldsymbol{\sigma} = \begin{bmatrix} \sigma_x & \sigma_y & \tau_{xy} \end{bmatrix}^T \tag{2}$$

and hence we say that the body (wall, membrane) is in the plane stress state.

In the column vector $\boldsymbol{\varepsilon}$ of the strain components, we will also consider only three components (In fact material reacts to the plane stress state by the transverse deformation ε_z, but this can be determined from the deformations in the midplane.)

$$\boldsymbol{\varepsilon} = \begin{bmatrix} \varepsilon_x & \varepsilon_y & 2\varepsilon_{xy} \end{bmatrix} \tag{3}$$

After simplifications

$$\sigma_z = \tau_{yz} = \tau_{zx} = \varepsilon_{yz} = \varepsilon_{zx} = 0 \tag{4}$$

the strains in the plane (*x*,*y*) are given by

$$\varepsilon_x = \frac{\partial u}{\partial x}, \quad \varepsilon_y = \frac{\partial v}{\partial y}, \quad 2\varepsilon_{xy} = \frac{\partial u}{\partial y} + \frac{\partial v}{\partial x} \tag{5}$$

From the physical (material) equations we have

$$\varepsilon_x = \frac{1}{E}\left(\sigma_x - \nu\sigma_y\right), \qquad \varepsilon_y = \frac{1}{E}\left(-\nu\sigma_x + \sigma_y\right), \qquad 2\varepsilon_{xy} = \frac{2(1+\nu)}{E}\tau_{xy} = \frac{\tau_{xy}}{G} \tag{6}$$

and from the condition $\sigma_z = 0$

$$\varepsilon_z = -\frac{\nu}{E}\left(\sigma_x + \sigma_y\right) \tag{7}$$

Inverting (6) we get the stress components

$$\sigma_x = \frac{E}{1-\nu^2}\left(\varepsilon_x + \nu\varepsilon_y\right), \quad \sigma_y = \frac{E}{1-\mu^2}\left(\nu\varepsilon_x + \varepsilon_y\right), \quad \tau_{xy} = G\gamma_{xy} \tag{8}$$

Now the strain ε_z can be expressed by substitution (8) to (7)

$$\varepsilon_z = -\frac{\nu}{1-\nu}\left(\varepsilon_x + \varepsilon_y\right) \tag{9}$$

The physical (constitutive) relationships in the matrix form are given by

$$\boldsymbol{\sigma} = \mathbf{D}^e\boldsymbol{\varepsilon} = \frac{E}{1-\nu^2}\begin{bmatrix} 1 & \nu & 0 \\ \nu & 1 & 0 \\ 0 & 0 & (1-\nu)/2 \end{bmatrix}\begin{bmatrix} \varepsilon_x \\ \varepsilon_y \\ 2\varepsilon_{xy} \end{bmatrix} \tag{10}$$

In the equations above E is the Young modulus, ν denotes the Poisson ratio and

$$G = \frac{E}{2(1+\nu)} \tag{11}$$

is the shear modulus of the material.

2. Von Mises plane stress isotropic small strain plasticity with the linear hardening

In this case, we again consider only those strain components that lie in the midplane because again ε_z can be expressed by using these components and for the plastic strain ε_z^p we consider incompressible material at plastic yielding ($\nu = 0.5$). Thus, we have

$$\varepsilon_z = \varepsilon_z^e + \varepsilon_z^p = -\left[\frac{\nu}{1-\nu}(\varepsilon_x^e + \varepsilon_y^e) + \varepsilon_x^p + \varepsilon_y^p\right] \tag{12}$$

where from the assumption of small strain the split of ε_z into elastic and plastic part is introdusced.

We need the *von Mises equivalent stress* $\bar{\sigma}$ for the plasticity condition

$$f = \bar{\sigma} - \bar{\sigma}_k(\bar{\varepsilon}^p) = 0 \tag{13}$$

which at the plane stress state has form

$$\bar{\sigma} = \sqrt{\sigma_x^2 + \sigma_y^2 - \sigma_x\sigma_y + 3\tau_{xy}^2} \tag{14}$$

and where $\bar{\sigma}_k(\bar{\varepsilon}^p)$ is *the uniaxial hardened yield stress* – a function of the *accumulated plastic strain* $\bar{\varepsilon}^p$. Then *the vector of plastic yielding* is given by

$$\mathbf{f} = \left\{\frac{\partial f}{\partial \boldsymbol{\sigma}}\right\} = \frac{1}{2\bar{\sigma}}\begin{Bmatrix} 2\sigma_x - \sigma_y \\ 2\sigma_y - \sigma_x \\ 6\tau_{xy} \end{Bmatrix} \tag{15}$$

3. Stress increment specification

For the *elastic test (trial) stress* in the numerical incremental integration step from n to $n+1$ we have

$$\boldsymbol{\sigma}_{n+1}^{test} = \begin{bmatrix} \sigma_{0x} \\ \sigma_{0y} \\ \tau_{0xy} \end{bmatrix} = \begin{bmatrix} \sigma_{nx} \\ \sigma_{ny} \\ \tau_{nxy} \end{bmatrix} + \frac{E}{1-\nu^2} \begin{bmatrix} 1 & \nu & 0 \\ \nu & 1 & 0 \\ 0 & 0 & (1-\nu)/2 \end{bmatrix} \begin{bmatrix} \Delta\varepsilon_x \\ \Delta\varepsilon_y \\ 2\Delta\varepsilon_{xy} \end{bmatrix} = \boldsymbol{\sigma}_n + \mathbf{D}^e \Delta\boldsymbol{\varepsilon} \tag{16}$$

and the equations of stress components after the stress point projection onto the updated yield surface are

$$\boldsymbol{\sigma}_{n+1} = \begin{bmatrix} \sigma_x \\ \sigma_y \\ \tau_{xy} \end{bmatrix} = \boldsymbol{\sigma}_{n+1}^{test} - \Delta\lambda \mathbf{D}^e \mathbf{f}_{n+1} = \begin{bmatrix} \sigma_{0x} \\ \sigma_{0y} \\ \tau_{0xy} \end{bmatrix} - \Delta\lambda \frac{E}{1-\nu^2} \begin{bmatrix} 1 & \nu & 0 \\ \nu & 1 & 0 \\ 0 & 0 & (1-\nu)/2 \end{bmatrix} \frac{1}{2\bar{\sigma}} \begin{bmatrix} 2\sigma_x - \sigma_y \\ 2\sigma_y - \sigma_x \\ 6\tau_{xy} \end{bmatrix} \tag{17}$$

where $\Delta\lambda$ is the unknown *incremental plastic multiplier*.

After an arrangement of equations (17), we get

$$\begin{aligned} \sigma_x + \sigma_y &= \frac{\sigma_{0x} + \sigma_{0y}}{1 + K\Delta\bar{\lambda}} \\ \sigma_x - \sigma_y &= \frac{\sigma_{0x} - \sigma_{0y}}{1 + 3G\Delta\bar{\lambda}} \\ \tau_{xy} &= \frac{\tau_{0xy}}{1 + 3G\Delta\bar{\lambda}} \end{aligned} \tag{18}$$

where

$$K = \frac{E}{2(1-\nu)}, \qquad \Delta\bar{\lambda} = \frac{\Delta\lambda}{\bar{\sigma}}$$

The unknown plastic multiplier $\Delta\bar{\lambda}$ has to satisfy the plasticity condition (13)

$$f_{n+1}(\Delta\bar{\lambda}) = \bar{\sigma}_{n+1}(\Delta\bar{\lambda}) - \bar{\sigma}_k(\Delta\bar{\lambda}) = 0 \tag{19}$$

where for a linear hardening material the hardened yield stress $\bar{\sigma}_k$ is

$$\bar{\sigma}_k(\Delta\bar{\lambda}) = \sigma_k + H\bar{\varepsilon}_p = \sigma_k + H\Delta\lambda = \sigma_k + H\bar{\sigma}_{n+1}\Delta\bar{\lambda} \tag{20}$$

depending on the *initial yield stress* σ_k, the accumulated plastic strian $\bar{\varepsilon}_p$, and the *hardening modulus*

$$H = \frac{EE_t}{E - E_t} \tag{21}$$

where E_t is so-called the *elastoplastic tangent modulus*.

The relationship for the $\Delta\lambda$ determitation has a more simple form when one uses the square of the condition (19)

$$q(\Delta\bar{\lambda}) = f_{n+1}^2(\Delta\bar{\lambda}) = \bar{\sigma}_{n+1}^2(\Delta\bar{\lambda}) - \bar{\sigma}_k^2(\Delta\bar{\lambda}) = 0 \tag{22}$$

Squaring the von Mises stress $\bar{\sigma}$ (14) we get

$$\bar{\sigma}_{n+1}^2 = \frac{1}{4}\left[(\sigma_x + \sigma_y)^2 + 3(\sigma_x - \sigma_y)^2 + 12\tau_{xy}^2\right] \tag{23}$$

and after substituting (18)

$$\bar{\sigma}_{n+1}^2 = \frac{(\sigma_{0x} + \sigma_{0y})^2}{4(1 + K\Delta\bar{\lambda})^2} + \frac{3(\sigma_{0x} - \sigma_{0y})^2 + 12\tau_{0xy}^2}{4(1 + 3G\Delta\bar{\lambda})^2} \tag{24}$$

Then the auxiliary function q (22) is given by

$$q(\Delta\bar{\lambda}) = \frac{(\sigma_{0x} + \sigma_{0y})^2}{4(1 + K\Delta\bar{\lambda})^2} + \frac{3(\sigma_{0x} - \sigma_{0y})^2 + 12\tau_{0xy}^2}{4(1 + 3G\Delta\bar{\lambda})^2} - \bar{\sigma}_k^2(\Delta\bar{\lambda}) = 0 \tag{25}$$

The nonlinear equation (25), in fact, the other form of the consistency condition (13), enables us to solve $\Delta\bar{\lambda}$ by the Newton-Raphson iteration. Decomposition of the function q in the reduced Tylor series gives the iterative relationship

$$\Delta\bar{\lambda}_{k+1} = \Delta\bar{\lambda}_k - \frac{q\left(\Delta\bar{\lambda}_k\right)}{q'\left(\Delta\bar{\lambda}_k\right)} \tag{26}$$

where q' is the derivative of function q by $\Delta\bar{\lambda}$ and which form according to (25) and (20) is as follows

$$q'\left(\Delta\bar{\lambda}\right) = \frac{\partial q}{\partial \Delta\bar{\lambda}} = 2\bar{\sigma}_{n+1}\frac{\partial \bar{\sigma}_{n+1}}{\partial \Delta\bar{\lambda}} - 2\bar{\sigma}_k\frac{\partial \bar{\sigma}_k}{\partial \Delta\bar{\lambda}} = 2\bar{\sigma}_{n+1}\frac{\partial \bar{\sigma}_{n+1}}{\partial \Delta\bar{\lambda}} - 2H\left(\sigma_k + H\bar{\sigma}_{n+1}\Delta\bar{\lambda}\right)\left(\bar{\sigma}_{n+1} + \Delta\bar{\lambda}\frac{\partial \bar{\sigma}_{n+1}}{\partial \Delta\bar{\lambda}}\right) \tag{27}$$

From (24) after extraction and differentiation by $\Delta\bar{\lambda}$ we get the last needed relationship for the iterative procedure

$$\frac{\partial \bar{\sigma}_{n+1}}{\partial \Delta\bar{\lambda}} = -\frac{1}{\bar{\sigma}_{n+1}}\left(\frac{K\left(\sigma_{0x} + \sigma_{0y}\right)^2}{4\left(1 + K\Delta\bar{\lambda}\right)^3} + \frac{3G\left[\left(\sigma_{0x} + \sigma_{0y}\right)^2 + 12\tau_{0xy}^2\right]}{4\left(1 + 3G\Delta\bar{\lambda}\right)^3}\right) \tag{28}$$

The calculation of $\Delta\bar{\lambda}$ and the resulting stress $\boldsymbol{\sigma}_{n+1}$ shows the following example (numerical results were calculated by the *Mathematica 5* program given in Fig. 2).

Example 1

Consider a body made of steel material and loaded by plane stress. Its properties are approximated by von Mises loading function and linear hardening. Consider a material particle, which previous state is without stress, deformed by strains $\varepsilon_x = 0{,}002$, $\varepsilon_y = -0{,}001$ and $\gamma_{xy} = 2\varepsilon_{xy} = 0{,}002$ and determine from the explicit relationships given above the final values of the stress components in the next integration step, i.e. $\boldsymbol{\sigma}_{n+1}$. Given is E = 200000 MPa, σ_k = 200 MPa, the linear elastoplastic tangent modulus $E_t = 100000$ MPa, ν = 0,3. The required accuracy is $f_{n+1} = \bar{\sigma}_{n+1} - \bar{\sigma}_k < 0{,}001$ MPa.

Solution

The test stress (16) is

$$\boldsymbol{\sigma}_0 = \begin{bmatrix} \sigma_{0x} \\ \sigma_{0y} \\ \tau_{0xy} \end{bmatrix} = \begin{bmatrix} 0 \\ 0 \\ 0 \end{bmatrix} + \frac{E}{1-\mu^2}\begin{bmatrix} 1 & \mu & 0 \\ \mu & 1 & 0 \\ 0 & 0 & (1-\mu)/2 \end{bmatrix}\begin{bmatrix} \Delta\varepsilon_x \\ \Delta\varepsilon_y \\ \Delta\gamma_{xy} \end{bmatrix} = \mathrm{L} = \begin{bmatrix} 373{,}6 \\ -87{,}9 \\ 153{,}8 \end{bmatrix} \mathrm{MPa}$$

Test of plasticity condition (13) gives

$$\bar{\sigma}_0 = \sqrt{\sigma_{0x}{}^2 + \sigma_{0y}{}^2 - \sigma_{0x}\sigma_{0y} + 3\tau_{0xy}^2} = \sqrt{373{,}6^2 + (-87{,}9)^2 - 373{,}6(-87{,}9) + 3\cdot 153{,}8^2} =$$

$$= 501{,}1 \text{ MPa} > \sigma_k = 200 \text{ MPa}$$

Because the test elastic stress $\bar{\sigma}_0$ is greater than the initial yield stress σ_k of the material point in the investigation, it follows that there will occur an elastic-plastic deformation. The resulting stress in such case has to be determined from the above backward integration relationships.

The initial values for calculation of stress at the elastic-plastic loading are $\Delta\bar{\lambda}_0 = 0$ and $\bar{\sigma}_{k0} = \sigma_k = 200$. Then, according to (24) up to (28) the iterative calculation was done by the program (the used relations do not contain matrices, and so we have chosen program *Mathematica 5*) shown in Fig. 2 and the results are:

The components of the resulting stress in the investigated point of the body:

$$\sigma_x = 266{,}0 \text{ MPa}$$

$$\sigma_y = -45{,}8 \text{ MPa}$$

$$\tau_{xy} = 103{,}9 \text{ MPa}$$

The resulting (hardened) yield stress of the material:

$$\overline{\sigma}_{kn+1} = 342{,}7 \text{ MPa}$$

The resulting equivalent plastic strain:

$$\overline{\varepsilon}_p = 7{,}135\ 10^{-4}$$

```
(* Backward-Euler, Plane stress - explicitly *)
Off[General::spell,General::spell1]

(* Input values *)
EE=200000; mi=0.3; Sigk0=200; G=EE/(2+2*mi); Et=100000; H=EE*Et/(EE-Et);

(* Start point *)
sxn=0;
syn=0;
sxyn=0;

(* Deformation increment *)
EpsX= 0.002;
EpsY=-0.001;
EpsXY=0.002;

(* Elastic trial stress *)
K=EE/2/(1-mi);
K1=EE/(1-mi^2);
sx0=sxn+K1*(EpsX+mi*EpsY);
sy0=syn+K1*(EpsY+mi*EpsX);
sxy0=sxyn+K1/2*(1-mi)*EpsXY;

(* Initial values of iteration *)
La=0;
Sigk=Sigk0;
Sekv=0;
c1=(sx0+sy0)^2;
c2=3*(sx0-sy0)^2+12*sxy0^2;

(* Iterative correction *)
While [Abs[Sigk-Sekv]>0.001,
Sekv2=0.25*(c1/(1+K*La)^2+c2/(1+3*La*G)^2);
q=Sekv2-Sigk^2;
Sekv=Sqrt[Sekv2];
dSekv=-1/(Sekv)*(K*c1/(1+K*La)^3+3*G*c2/(1+3*G*La)^3);
dq=2*Sekv*dSekv-2*Sigk*(H*Sekv+La*H*dSekv);
La=La-q/dq;
Sigk=Sigk0+H*La*Sekv
];

(* Stess calculation *)
A1=1+0.5*La*EE/(1-mi);
A2=1+3*La*G;
sps=(sx0+sy0)/A1;
sms=(sx0-sy0)/A2;
SXdef=(sps+sms)/2;
SYdef=(sps-sms)/2;
SXYdef=sxy0/A2;

(* Value of the new yield stress *)
SigkDef=Sigk;

(* Equivalent plastic deformation *)
EplastPruh=La*Sekv;

Print["Výsledky"/TableForm[{{"SX =", SXdef,"MPa"},
                 {"SY =", SYdef,"MPa"},
                 {"SXY =", SXYdef,"MPa"},
                 {"SigK =", SigkDef,"MPa"},
                 {"EpsPlast =", EplastPruh}
                 }]]
```

Výsledky

SX =	265.994	MPa
SY =	−45.7719	MPa
SXY =	103.922	MPa
SigK =	342.669	MPa
EpsPlast =	0.000713346	

Fig. 2 *Mathematica 5 program and results of the example 1 solution (Výsledky = Results)*

Example 2

Solve the previous example by Ansys.

Solution

Material particle of the body can be replaced by a single plane stress finite element of unit thickness with given material properties. The problem of the γ_{xy} input can be solved by placing the element edges in the directions of the main strains ε_1 and ε_2 for which we have

$$\varepsilon_{1,2} = \frac{\varepsilon_x + \varepsilon_y}{2} \pm \frac{1}{2}\sqrt{\left(\varepsilon_x - \varepsilon_y\right)^2 + \left(\gamma_{xy}\right)^2} = \frac{0,002 - 0,001}{2} \pm \frac{1}{2}\sqrt{\left(0,002 + 0,001\right)^2 + \left(0,002\right)^2} =$$
$$= 0,0005 \pm 0,0018027 \quad \rightarrow \quad \varepsilon_1 = 0,0023027; \quad \varepsilon_2 = -0,0013027$$

and for the direction of ε_1 we receive

$$tg2\varphi = \frac{\gamma_{xy}}{\varepsilon_x - \varepsilon_y} = \frac{0,002}{0,002 + 0,001} = \frac{2}{3} \quad \rightarrow \quad \varphi = 16,845°$$

Now the example can be solved in the Ansys interactive environment by these steps:

1. Job name
Utility Menu>File>Change Jobname..., /FILNAM = PlaneStress1, OK;

2. Element type
Main Menu>Preprocessor>Element Type>Add/Edit/Delete, Add..., Solid Quad 8 node 183, OK, Close;

3. Material properties
Linear
Preprocessor>Material Props>Material Models, Structural, Linear, Elastic, Isotropic, EX = 2E5, PRXY = 0.3, OK,
Nonlinear
Nonlinear, Inelastic, Rate Independent, Isotropic Hardening Plasticity, Mises Plasticity, Bilinear, Yield Strss = 200, Tang Mod = 1E5, OK, Material, Exit;

4. Element keypoints (numbering is automatic)
Utility Menu>Work Plane>Offset WP by Increments..., Degrees = 16.845, 0, 0, OK;
Utility Menu>Work Plane>Local Coordinate Systems>Create Local CS>At WP Origin..., OK;
Preprocessor>Modeling>Create>Keypoints>In Active CS: X = 0, Y = 0, Apply,
X = 1, Y = 0, Apply,
X = 1, Y = 1, Apply,
X = 0, Y = 1, OK;

5. Element area
Preprocessor>Modeling>Create>Areas>Arbitrary>Trough KPs: ↑KP1,↑KP2,↑KP3,↑KP4, OK;

6. Element creating
Preprocessor>Meshing>Mesh Tool: Size Controls: Lines, Set: Pick All, NDIV = 1, OK;
(Preprocessor>Meshing>Mesh Tool): Mesh, Pick All, Close;

7. ε_1 and ε_2 input
Preprocessor>Modeling>Move/Modify>Rotate Node CS>To Active CS ↑, Pick All;
Main Menu>Solution>Define Loads>Apply>Structural>Displacement>On Nodes

List of Items: 1, 8, 6, OK, UX, Value = 0, Apply,
1, 3, 2, OK, UY, Value = 0, Apply,
2, 5, 4, OK, UX, Value = 0.0023027, Apply,
6, 7, 4, OK, UY, Value = - 0.0013027, OK;

8. Solution

*Main Menu>Solution>Analysis Type>Sol'n Controls...,*Time at end of loadstep = 1, Number of substeps = 1, OK;

Solution>Solve>Current LS, Solve Current Load Step, OK;



9. Stress results

Main Menu>General Postproc>List Results> Nodal Solution..., Stress, X-Component of stress, OK;

```
 PRINT S    NODAL SOLUTION PER NODE
  THE FOLLOWING X,Y,Z VALUES ARE IN GLOBAL COORDINATES

    NODE     SX           SY           SZ           SXY          SYZ          SXZ
       1   265.99     -45.766      0.0000       103.92       0.0000       0.0000
       2   265.99     -45.766      0.0000       103.92       0.0000       0.0000
       4   265.99     -45.766      0.0000       103.92       0.0000       0.0000
       6   265.99     -45.766      0.0000       103.92       0.0000       0.0000
```

10. End of job

*Ansys Toolbar>Quit>*Save Geom+Loads, OK;

The results are the same as at the analytical solution.

4. <u>Matrix formulation</u>

The mentioned analytical determination of stress state in the elastic-plastic load step is indeed an elegant and simple procedure but if we are interested in creating a FEM program for solving elastic-plastic plane stress problems it is necessary to establish basic relations and save the necessary values for the next load step in the matrix form. In such case, the state variables are written to the column matrices (vectors)

$$\boldsymbol{\varepsilon} = \left[\varepsilon_x, \varepsilon_y, 2\varepsilon_{xy}\right]^T, \quad \boldsymbol{\varepsilon}^e = \left[\varepsilon_x^e, \varepsilon_y^e, 2\varepsilon_{xy}^e\right]^T, \quad \boldsymbol{\varepsilon}^p = \left[\varepsilon_x^p, \varepsilon_y^p, 2\varepsilon_{xy}^p\right]^T$$
$$\boldsymbol{\sigma} = \left[\sigma_x, \sigma_y, \tau_{xy}\right]^T, \quad \mathbf{s} = \left[s_x, s_y, s_{xy}\right]^T \tag{29}$$

where at the deformation variables we consider tensor components for shear strain ($\gamma_{xy} = 2\varepsilon_{xy}$) and for the deviatoric stress component we will implement

$$\mathbf{s} = \mathbf{P}\boldsymbol{\sigma} \tag{30}$$

where

$$\mathbf{P} = \mathbf{P}^T = \begin{bmatrix} \frac{2}{3} & -\frac{1}{3} & 0 \\ -\frac{1}{3} & \frac{2}{3} & 0 \\ 0 & 0 & 2 \end{bmatrix} \tag{31}$$

Now we can express the von Mises equivalent stress (14) as follows

$$\bar{\sigma}=\sqrt{\sigma_x^2+\sigma_y^2-\sigma_x\sigma_y+3\tau_{xy}}=\sqrt{\tfrac{3}{2}\mathbf{s}^T\boldsymbol{\sigma}}=\sqrt{\tfrac{3}{2}\boldsymbol{\sigma}^T\mathbf{P}\boldsymbol{\sigma}} \tag{32}$$

The plasticity condition (13) is given by

$$f=\sqrt{\tfrac{3}{2}\boldsymbol{\sigma}^T\mathbf{P}\boldsymbol{\sigma}}-\bar{\sigma}_k(\bar{\varepsilon}^p)=0 \tag{33}$$

and its preferred squared form is

$$f_2=\tfrac{1}{2}\boldsymbol{\sigma}^T\mathbf{P}\boldsymbol{\sigma}-\tfrac{1}{3}\bar{\sigma}_k^2(\bar{\varepsilon}^p)=0 \tag{34}$$

The easiest way to get the increment of the plastic deformation vector is by using the formula for deriving the scalar quadratic form

$$\Delta\boldsymbol{\varepsilon}^p=\Delta\lambda\frac{\partial f_2}{\partial\boldsymbol{\sigma}}=\Delta\lambda\frac{\partial}{\partial\boldsymbol{\sigma}}(\tfrac{1}{2}\boldsymbol{\sigma}^T\mathbf{P}\boldsymbol{\sigma})=\Delta\lambda\mathbf{P}\boldsymbol{\sigma} \tag{35}$$

where for the vector of plasticity in this case we have

$$\mathbf{f}=\frac{\partial f_2}{\partial\boldsymbol{\sigma}}=\mathbf{P}\boldsymbol{\sigma}=\mathbf{s} \tag{36}$$

The increment of plastic deformation is

$$\Delta\bar{\varepsilon}^p=\Delta\lambda\sqrt{\tfrac{2}{3}\mathbf{s}^T\boldsymbol{\sigma}}=\Delta\lambda\sqrt{\tfrac{2}{3}\boldsymbol{\sigma}^T\mathbf{P}\boldsymbol{\sigma}} \tag{37}$$

In an increment step from time t_n to time t_{n+1}, the values of the elastic predictor are

$$\begin{aligned}\boldsymbol{\varepsilon}_{n+1}^{e\,test}&=\boldsymbol{\varepsilon}_n^e+\Delta\boldsymbol{\varepsilon}\\ \boldsymbol{\sigma}_{n+1}^{test}&=\mathbf{D}^e\boldsymbol{\varepsilon}_{n+1}^{e\,test}\\ \bar{\boldsymbol{\varepsilon}}_{n+1}^{p\,test}&=\bar{\boldsymbol{\varepsilon}}_n^p\end{aligned} \tag{38}$$

where $\Delta\boldsymbol{\varepsilon}$ is the known linear strain increment calculated from the load increment. In (38) all values are known and we can check the plasticity condition (34). First, we calculate

$$f_2^{test}=\tfrac{1}{2}(\boldsymbol{\sigma}_{n+1}^{test})^T\mathbf{P}\boldsymbol{\sigma}_{n+1}^{test}-\tfrac{1}{3}\bar{\sigma}_k^2(\bar{\varepsilon}_{n+1}^{test}) \tag{39}$$

and if $f_2^{test}\le 0$, then the test values are valid for this (elastic) load step, if this is not the case, then we have to return these values to the changed theplasticity area by the plastic corrector.

5. <u>Equations for the return of the stress point on the yield area (plastic corrector)</u>

The first equation, which the unknow stress $\boldsymbol{\sigma}_{n+1}$ have to fulfill at a plastic load step, is the condition (34) at the time t_{n+1}

$$f_2=\tfrac{1}{2}\boldsymbol{\sigma}_{n+1}^T\mathbf{P}\boldsymbol{\sigma}_{n+1}-\tfrac{1}{3}\bar{\sigma}_k^2(\bar{\varepsilon}_{n+1})=0 \tag{40}$$

The equivalent plastic deformation in the load step will change by the increment (37) to

$$\bar{\varepsilon}_{n+1}^p=\bar{\varepsilon}_n^p+\Delta\lambda\sqrt{\tfrac{2}{3}\boldsymbol{\sigma}_{n+1}^T\mathbf{P}\boldsymbol{\sigma}_{n+1}} \tag{41}$$

and the last equation in this system of nonlinear equations is the implicit relationship of the backward Euler method for calculation of the elastic deformation

$$\boldsymbol{\varepsilon}_{n+1}^e=\boldsymbol{\varepsilon}_{n+1}^{etest}-\Delta\lambda\mathbf{f}_{n+1}=\boldsymbol{\varepsilon}_{n+1}^{etest}-\Delta\lambda\mathbf{P}\boldsymbol{\sigma}_{n+1} \tag{42}$$

After a suitable arrangement the system of equations (40) to (42) can be changed to only one nonlinear equation with only one unknown $\Delta\lambda$. Startin with the elastic constitutive equations [1]

$$\boldsymbol{\sigma}_{n+1}=\mathbf{D}^e\boldsymbol{\varepsilon}_{n+1}^e=\mathbf{D}^e(\boldsymbol{\varepsilon}_{n+1}^{etest}-\Delta\lambda\mathbf{P}\boldsymbol{\sigma}_{n+1})=\mathbf{D}^e(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_n^p-\Delta\lambda\mathbf{P}\boldsymbol{\sigma}_{n+1}) \tag{43}$$

gradually we get

$$
\begin{aligned}
(\mathbf{I}+\Delta\lambda\mathbf{D}^{e}\mathbf{P})\boldsymbol{\sigma}_{n+1} &= \mathbf{D}^{e}(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_{n}^{p}) \\
\boldsymbol{\sigma}_{n+1} &= (\mathbf{I}+\Delta\lambda\mathbf{D}^{e}\mathbf{P})^{-1}\mathbf{D}^{e}(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_{n}^{p}) \\
\boldsymbol{\sigma}_{n+1} &= (\mathbf{D}^{e-1}+\Delta\lambda\mathbf{P})^{-1}(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_{n}^{p})
\end{aligned}
\tag{44}
$$

With the known $\Delta\lambda$ it is possible, according to (44), to calculate the stress components from

$$\boldsymbol{\sigma}_{n+1}=\hat{\mathbf{D}}(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_{n}^{p}) \tag{45}$$

or, because $\boldsymbol{\sigma}_{n+1}^{test}=\mathbf{D}^{e}(\boldsymbol{\varepsilon}_{n+1}-\boldsymbol{\varepsilon}_{n}^{p})$, from

$$\boldsymbol{\sigma}_{n+1}=\hat{\mathbf{D}}\mathbf{D}^{e-1}\boldsymbol{\sigma}_{n+1}^{test} \tag{46}$$

where

$$\hat{\mathbf{D}}=(\mathbf{D}^{e-1}+\Delta\lambda\mathbf{P})^{-1} \tag{47}$$

6. <u>Specifying the plastic multiplier from the consistency condition</u>

There is only one unknown in the above equations, namely the plastic multiplier $\Delta\lambda$. Its value has to fulfill the consistency condition (40). Direct substitution of the matrix relations for $\boldsymbol{\sigma}_{n+1}$ (46) in this condition leads to an unpleasant matrix form of a nonlinear equation and therefore is better to use an explicit form which derivation we will show in this section. In the case of isotropic elastic material the matrices $\mathbf{D}^{e}$ and $\mathbf{P}$ have the same eigenvectors and it is possible by their orthogonal transformation in the plasticity condition to change them to diagonal and so to get an explicit equation form suitable for $\Delta\lambda$ calculation. The transformation relations are

$$\mathbf{P}=\mathbf{Q}\boldsymbol{\Lambda}_{\mathrm{P}}\mathbf{Q}^{T};\qquad \mathbf{D}^{e}=\mathbf{Q}\boldsymbol{\Lambda}_{\mathrm{D}}\mathbf{Q}^{T};\qquad \mathbf{Q}=\frac{1}{\sqrt{2}}\begin{bmatrix}1 & -1 & 0\\ 1 & 1 & 0\\ 0 & 0 & \sqrt{2}\end{bmatrix};\qquad \mathbf{Q}^{-1}=\mathbf{Q}^{T} \tag{48}$$

where

$$\boldsymbol{\Lambda}_{\mathrm{P}}=\begin{bmatrix}1/3 & 0 & 0\\ 0 & 1 & 0\\ 0 & 0 & 2\end{bmatrix};\qquad \boldsymbol{\Lambda}_{\mathrm{D}}=\begin{bmatrix}E/(1-\mu) & 0 & 0\\ 0 & 2G & 0\\ 0 & 0 & G\end{bmatrix} \tag{49}$$

and $\mathbf{D}^{e}$ is given by (10).

From (44) we get

$$\boldsymbol{\sigma}_{n+1}=(\mathbf{I}+\Delta\lambda\mathbf{D}^{e}\mathbf{P})^{-1}\boldsymbol{\sigma}_{n+1}^{test}=\mathbf{Q}\boldsymbol{\Gamma}\mathbf{Q}^{T}\boldsymbol{\sigma}_{n+1}^{test} \tag{50}$$

where

$$\boldsymbol{\Gamma}=(\mathbf{I}+\boldsymbol{\Lambda}_{\mathrm{D}}\boldsymbol{\Lambda}_{\mathrm{P}})^{-1}=\begin{bmatrix}\left(1+\frac{\Delta\lambda E}{3(1-\mu)}\right)^{-1} & 0 & 0\\ 0 & (1+2\Delta\lambda G)^{-1} & 0\\ 0 & 0 & (1+2\Delta\lambda G)^{-1}\end{bmatrix} \tag{51}$$

Substituting (50) into the consistency condition (40) gives the simplified explicit equation

$$\tfrac{1}{2}\left(\boldsymbol{\sigma}_{n+1}^{test}\right)^{T}\mathbf{Q}\boldsymbol{\Gamma}\boldsymbol{\Lambda}_{\mathrm{P}}\boldsymbol{\Gamma}\mathbf{Q}^{T}\boldsymbol{\sigma}_{n+1}^{test}-\tfrac{1}{3}\overline{\sigma}_{k}^{2}(\overline{\varepsilon}_{n+1}^{p})=\tfrac{1}{2}\phi^{2}-\tfrac{1}{3}S^{2}=0 \tag{52}$$

where

$$\phi^{2}=\frac{A}{(1+a\Delta\lambda)^{2}}+\frac{B}{(1+b\Delta\lambda)^{2}}+\frac{C}{(1+b\Delta\lambda)^{2}} \tag{53}$$

with

$$A=\tfrac{1}{6}\left(\sigma_x^{test}+\sigma_y^{test}\right)^2;\quad B=\tfrac{1}{2}\left(\sigma_x^{test}-\sigma_y^{test}\right)^2;\quad C=2\left(\tau_{xy}^{test}\right)^2$$
$$a=\tfrac{1}{3}\Delta\lambda E/\left(1-\mu\right);\quad b=2G$$

For the yield stress at the end of the load step we have

$$\bar{\sigma}_k^2(\bar{\varepsilon}_{n+1}^p)=S^2=\sigma_k+H\left(\bar{\varepsilon}_n^p+\Delta\lambda\sqrt{\tfrac{2}{3}}\phi\right) \tag{54}$$

Now the last problem is to calculate $\Delta\lambda$ from the nonlinear equation (52) .

7. <u>Determination of the plastic multiplier by the Newton-Raphson method</u>

We are searching such value of the plastic multiplier at which the equation (52) will be fulfilled with sufficient accuracy. In other words at the Newton-Raphson method the value of $\Delta\lambda$ has to change in such way that function

$$q\left(\Delta\lambda\right)=\tfrac{1}{2}\phi^2-\tfrac{1}{3}S^2 \tag{55}$$

converges to zero. The classical iterative formula for an equation with one unknown in our case gives

$$\Delta\lambda^{k+1}=\Delta\lambda^k-\frac{q\left(\Delta\lambda^k\right)}{q'\left(\Delta\lambda^k\right)} \tag{56}$$

where

$$q'\left(\Delta\lambda^k\right)=\frac{\partial q}{\partial\Delta\lambda}=\phi\frac{\partial\phi}{\partial\Delta\lambda}-\frac{2}{3}S\frac{\partial S}{\partial\Delta\lambda} \tag{57}$$

The partial derivatives needed for q' can be determined from (53) and (54)

$$\frac{\partial\phi}{\partial\Delta\lambda}=-\frac{1}{2\phi}\left(\frac{2Aa}{\left(1+a\Delta\lambda\right)^3}+\frac{2Bb}{\left(1+b\Delta\lambda\right)^3}+\frac{2Cb}{\left(1+b\Delta\lambda\right)^3}\right) \tag{58}$$

$$\frac{\partial S}{\partial\Delta\lambda}=H\sqrt{\tfrac{2}{3}}\left(\phi+\Delta\lambda\frac{\partial\phi}{\partial\Delta\lambda}\right) \tag{59}$$

8. <u>Matrix relations in Fortran</u>

In the FEM programs above relationships are written in the form of a subroutine which is called in the cycle for each integration element point of the body model. To conclude this section we show such subroutine in a form an executable program. The program is based on the above relations and it should be comprehensible for peaple with an elementary knowledge of Fortran language. The program solve the same example as was above solved with other programs (example 1 and 2).

```
      PROGRAM RETURN
C*****************************************************
C     Stress point return on the plasticity area    *
C*****************************************************
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION
     1    STRES(3) ,SVECT(3), EPS(3),
     2    A(3,3),B(3,3),P(3,3),D(3,3)
      DATA R0    ,R1    ,R2    ,R3    ,R4    ,R6
     1    /0.D0 ,1.D0 ,2.D0 ,3.D0 ,4.D0 ,6.D0 /
      DATA ALLOW / 0.00001D0 /
 100  FORMAT(/' The maximum number of iterations was exceeded/)
 200  FORMAT(' Medza sklzu =',F10.2)
 300  FORMAT(' Napatia SX, SY, SXY =',3F10.2)
      ROOT3 = SQRT(R3)
      DLAMBDA = R0
      I = 0
      EPEKV = 0.0D0
      E = 2.0D5
      POIS = 0.3D0
      F = E/(R3*(R1-POIS))
      G = E/(R2*(R1+POIS))
      CALL MATICA(D) !Matrix De calculation
      EPS(1) = 2.0D-3 !Given components of the deformation increment
      EPS(2) = -1.0D-3
```

```
      EPS(3) = 2.0D-3
      STRES(1) = D(1,1)*EPS(1)+D(1,2)*EPS(2) !Components of the trial stress
      STRES(2) = D(2,1)*EPS(1)+D(2,2)*EPS(2)
      STRES(3) = D(3,3)*EPS(3)
      H = 2.0D5        !the hardening modulus
      SIGK0 = 2.0D2 !the initial yield stress
      A1 = (STRES(1)+STRES(2))*(STRES(1)+STRES(2))
      A2 = (STRES(1)-STRES(2))*(STRES(1)-STRES(2))
      A3 = R2*STRES(3)*STRES(3)
      DLAMBDA = R0
      EPSTN = EPEKV
      SQ2D3 = SQRT(R2/R3)
      I = 0
  210 I = I+1  !Iteration cycle of the DeltaLambda calculation by N-R metod
      D1 = R1+DLAMBDA*F
      D2 = R1+R2*DLAMBDA*G
      FI = A1/(R6*D1*D1)+A2/(R2*D2*D2)+A3/(D2*D2)
      FI = SQRT(FI)
      DFI = -R2*A1*F/(R6*D1*D1*D1)-R4*A2*G/(R2*D2*D2*D2)
     1      -R4*A3*G/(D2*D2*D2)
      DFI = DFI/(R2*FI)
      EPEKV = EPSTN+DLAMBDA*FI*SQRT(R2/R3)
      SIGK = SIGK0+H*EPEKV
      DSIGK = H*SQ2D3*(FI+DLAMBDA*DFI)
      GFCIA = FI*FI/R2-SIGK*SIGK/R3
      DGFCIE = FI*DFI-R2*SIGK*DSIGK/R3
      DLAMBDA = DLAMBDA-GFCIA/DGFCIE
      IF(GFCIA.GT.ALLOW) GOTO 210
      IF(I.GT.50)write(*,100)
C
      P(1,1) = R2/R3   !P matrix
      P(1,2) = -R1/R3
      P(1,3) = R0
      P(2,1) = -R1/R3
      P(2,2) = R2/R3
      P(2,3) = R0
      P(3,1) = R0
      P(3,2) = R0
      P(3,3) = R2
      CALL INVERT(D,A)
      DO 220 I = 1,3
      DO 220 J = 1,3
      D(I,J) = A(I,J)+DLAMBDA*P(I,J)
  220 CONTINUE
      CALL INVERT(D,B)
      DO 230 I = 1,3
      DO 230 J = 1,3
      D(I,J) = R0
      DO 230 K = 1,3
      D(I,J) = D(I,J)+B(I,K)*A(K,J)
  230 CONTINUE
      DO 240 I = 1,3  !Stress calculation
      SVECT(I) = R0
      DO 240 J = 1,3
      SVECT(I) = SVECT(I)+D(I,J)*STRES(J)
  240 CONTINUE
      DO 250 I = 1,3
      STRES(I) = SVECT(I)
  250 CONTINUE
      STRES(4) = R0
C     Printing results on the monitor screen
       write(*,300)stres(1),stres(2),stres(3)
       write(*,200)SIGK  !Yield stress
       STOP
      END

      SUBROUTINE MATICA(D)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)

      DIMENSION  D(3,3)
      DATA R0,R1,R2/0.0D0,1.0D0,2.0D0/
      E = 2.0D5
      POIS = 0.3D0
      DO 10 ISTRE=1,3
      DO 10 JSTRE=1,3
   10 D(ISTRE,JSTRE)=R0
      CONST=E/(R1-POIS*POIS)
      D(1,1)=CONST
      D(2,2)=CONST
      D(1,2)=CONST*POIS
      D(2,1)=CONST*POIS
      D(3,3)=(R1-POIS)*CONST/R2
      RETURN
      END

      SUBROUTINE INVERT(A,B)
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION A(3,3),B(3,3)
      A11 = A(1,1)
      A12 = A(1,2)
      A13 = A(1,3)
      A21 = A(2,1)
      A22 = A(2,2)
      A23 = A(2,3)
```

```
      A31 = A(3,1)
      A32 = A(3,2)
      A33 = A(3,3)
      T1  = A22*A33 - A32*A23
      T2  = A23*A31 - A21*A33
      T3  = A21*A32 - A22*A31
      DETER = A11*T1 + A12*T2 + A13*T3
      IF(DETER.EQ.0.0) stop
      DENOM = 1./DETER
      B(1,1) = T1*DENOM
      B(2,1) = T2*DENOM
      B(3,1) = T3*DENOM
      B(1,2) = (-A12*A33 + A32*A13)*DENOM
      B(2,2) = ( A11*A33 - A31*A13)*DENOM
      B(3,2) = (-A11*A32 + A12*A31)*DENOM
      B(1,3) = ( A12*A23 - A13*A22)*DENOM
      B(2,3) = (-A11*A23 + A21*A13)*DENOM
      B(3,3) = ( A11*A22 - A21*A12)*DENOM
      RETURN
      END
```

Results (Napatia = Stresses, Medza sklzu = Yield stress):

```
Napatia SX, SY, SXY =     265.99    -45.77    103.92
Medza sklzu =     342.67
Stop - Program terminated.

Press any key to continue
```

9. Consistent tangent material modulus

The global tangential stiffness matrix of an FE model is created by an organized summation of its element tangential stiffness matrices

$$\mathbf{K}_T^e = \int_S \mathbf{B}^T \mathbf{D} \mathbf{B}\, dV \tag{60}$$

where at a material nonlinear problem the matrix $\mathbf{D}$ is so-colled *consistent tangential material modulus* of the element. The term consistent expresses a requirement, that its value has to be in every incremental step consistent with the integration procedure for the solution of the stress $\boldsymbol{\sigma}_{n+1}$. The definition of the tangential material modulus is

$$\mathbf{D} = \frac{\partial \boldsymbol{\sigma}_{n+1}}{\partial \boldsymbol{\varepsilon}_{n+1}} = \frac{\partial \boldsymbol{\sigma}_{n+1}}{\partial \boldsymbol{\varepsilon}_{n+1}^{e\,test}} \tag{61}$$

The derivations equality in (61) follows from validity

$$\boldsymbol{\varepsilon}_{n+1}^{e\,test} = \boldsymbol{\varepsilon}_{n+1} - \boldsymbol{\varepsilon}_n^p \tag{62}$$

The consistent linearization of the local material element matricies will produce the global tangential stiffness matrix in a such form, which will guarantee quadratic convergence of the global Newton-Raphson iteration in the load step.

The derivation of the $\mathbf{D}$ matrix begins with differentiation of expression for stress (43)

$$d\boldsymbol{\sigma}_{n+1} = d\mathbf{D}^e \boldsymbol{\varepsilon}_{n+1}^{test} - d\Delta\lambda \mathbf{D}^e \mathbf{P} \boldsymbol{\sigma}_{n+1} - \Delta\lambda \mathbf{D}^e \mathbf{P}\, d\boldsymbol{\sigma}_{n+1} \tag{63}$$

and after similar arrangements as at (43) we get

$$d\boldsymbol{\sigma}_{n+1} = \hat{\mathbf{D}}(d\boldsymbol{\varepsilon}_{n+1}^{test} - d\Delta\lambda \mathbf{P}\, d\boldsymbol{\sigma}_{n+1}) \tag{64}$$

It remains to express $d\Delta\lambda$ in (64) from the consistency condition.

Differentiation of the consistency condition (40) gives (note: $d(\mathbf{x}^T\mathbf{x}) = 2\mathbf{x}^T d\mathbf{x}$)

$$df_2 = \boldsymbol{\sigma}_{n+1}^T \mathbf{P}\, d\boldsymbol{\sigma}_{n+1} - \tfrac{2}{3}\bar{\sigma}_k H\, d\bar{\varepsilon}_{n+1}^p = 0 \tag{65}$$

From (41) we have

$$d\bar{\varepsilon}^p_{n+1} = d\Delta\lambda a + \frac{2}{3}\frac{\Delta\lambda}{a}\boldsymbol{\sigma}^T_{n+1}\mathbf{P}d\boldsymbol{\sigma}_{n+1} \tag{66}$$

where

$$a = (\tfrac{2}{3}\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\boldsymbol{\sigma}_{n+1})^{\frac{1}{2}} = \tfrac{2}{3}\bar{\sigma}_k$$

By substituting (66) in (65) we have

$$\boldsymbol{\sigma}^T_{n+1}\mathbf{P}d\boldsymbol{\sigma}_{n+1} - \tfrac{2}{3}\bar{\sigma}_k H(d\Delta\lambda a + \frac{2}{3}\frac{\Delta\lambda}{a}\boldsymbol{\sigma}^T_{n+1}\mathbf{P}d\boldsymbol{\sigma}_{n+1}) = 0 \tag{67}$$

and after substitute a we get a simplified equation

$$\boldsymbol{\sigma}^T_{n+1}\mathbf{P}d\boldsymbol{\sigma}_{n+1} - \frac{4}{9}\frac{H\bar{\sigma}_k^2}{b}d\Delta\lambda = 0 \tag{68}$$

where

$$b = 1 - \tfrac{2}{3}H\Delta\lambda$$

Substitution (64) into (68) gives the following relationship

$$\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}d\boldsymbol{\varepsilon}^{test}_{n+1} - d\Delta\lambda\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}\mathbf{P}\boldsymbol{\sigma}_{n+1} - \frac{4}{9}\frac{H\bar{\sigma}_k^2}{b}d\Delta\lambda = 0 \tag{69}$$

and from this equation the wanted differential is given by

$$d\Delta\lambda = \frac{\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}}{\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}\mathbf{P}\boldsymbol{\sigma}_{n+1} + \frac{4}{9}\frac{H\bar{\sigma}_k^2}{b}}d\boldsymbol{\varepsilon}^{test}_{n+1} \tag{70}$$

The use of $d\Delta\lambda$ in (64) gives the resulting expression for the stress increment

$$d\boldsymbol{\sigma}_{n+1} = \mathbf{D}d\boldsymbol{\varepsilon}^{test}_{n+1} \tag{71}$$

with the consistent tangential modulus

$$\mathbf{D} = \hat{\mathbf{D}} - \frac{\hat{\mathbf{D}}\mathbf{P}\boldsymbol{\sigma}_{n+1}\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}}{\boldsymbol{\sigma}^T_{n+1}\mathbf{P}\hat{\mathbf{D}}\mathbf{P}\boldsymbol{\sigma}_{n+1} + \frac{4}{9}\frac{H\bar{\sigma}_k^2}{b}} = \hat{\mathbf{D}} - \frac{\hat{\mathbf{D}}\mathbf{s}_{n+1}\mathbf{s}^T_{n+1}\hat{\mathbf{D}}}{\mathbf{s}^T_{n+1}\hat{\mathbf{D}}\mathbf{s}_{n+1} + \frac{4}{9}\frac{H\bar{\sigma}_k^2}{b}} \tag{72}$$

Thus, if a FEM program for solving the elastic-plastic plane stress problem use the incremental calculation of the stress according to equations (43) to (46), then to ensure the quadratic convergence of the Newson-Raphson method at solution of the global nonlinear system is necessary to use the material modulus (72) at the tangential element matrices (60) calculation. Use of the classical tangential material modulus (so-colled *elastoplastic continuum tangent operator*)

$$\mathbf{D}_{cont} = \mathbf{D}^e - \frac{9G^2}{\bar{\sigma}_k^2(3G+H)}\mathbf{s}\mathbf{s}^T \tag{73}$$

leads to a rapid decline inconvergence.

10. <u>The consistent material modulus computation in FORTRAN.</u>

According to the relationships of the previous section we have processed the program MATMODUL2D for the consistent tangential material modulus with linear isotropic hardening determination in Fortran. In a real FEM program, such part acts as a subroutine called for every integral point from the subroutine which calculates the tangential element stiffness matrices. The input variables come to such subroutine as parameters of the calling command but in this program illustration we have input them directly in the program part noted as *Input values*.

```
      PROGRAM MATMODUL2D
C*************************************************
C     CONSISTENT MATERIAL MODULUS VON MISES 2D   *
C*************************************************
      IMPLICIT DOUBLE PRECISION (A-H,O-Z)
      DIMENSION
     1    D(3,3),A(3,3),P(3,3),AVECT(3),BVECT(3),
     3    STRES(3),B(3,3)
      DATA R0      ,R1      ,R2      ,R3      ,R4      ,R9
     1    /0.0D0   ,1.0D0   ,2.0D0   ,3.0D0   ,4.0D0   ,9.0D0/
```

```
C Input values
      E=2.0D5
      H=2.0D5
      POIS=R0
      DLAMBDA=6.6851D-4
      STRES(1)=385.3D0
      STRES(2)=197.9D0
      STRES(3)=R0
      SIGEF=333.7D0
      SIGK=333.7D0
C
      BETA  = R1-H*DLAMBDA/(R3*SIGEF)
      DO 10 I=1,3
      DO 10 J=1,3
   10 D(I,J)=R0
C Elastic matrix De calculation
      CONST=E/(R1-POIS*POIS)
      D(1,1)=CONST
      D(2,2)=CONST
      D(1,2)=CONST*POIS
      D(2,1)=CONST*POIS
      D(3,3)=(R1-POIS)*CONST/R2
      D(4,4)=R1
C P-matrix
      P(1,1) = R2/R3
      P(1,2) = -R1/R3
      P(1,3) = R0
      P(2,1) = -R1/R3
      P(2,2) = R2/R3
      P(2,3) = R0
      P(3,1) = R0
      P(3,2) = R0
      P(3,3) = R2
C Expanded D-matrix
      CALL INVERT(D,A)
      DO 120 I=1,3
      DO 120 J=1,3
      B(I,J) = A(I,J)+DLAMBDA*P(I,J)
  120 CONTINUE
      CALL INVERT(B,A)
C Consistent tangential matrix calculation
      DO 140 I=1,3
      AVECT(I) = R0
      DO 140 J=1,3
      AVECT(I) = AVECT(I)+P(I,J)*STRES(J)
  140 CONTINUE
      DO 150 I=1,3
      BVECT(I) = R0
      DO 150 J=1,3
      BVECT(I) = BVECT(I)+A(I,J)*AVECT(J)
  150 CONTINUE
      RMENOV = R0
      DO 160 I=1,3
  160 RMENOV = RMENOV+AVECT(I)*BVECT(I)
      BETA = R1-R2*H*DLAMBDA/R3
      H2 = R4*H*SIGK*SIGK/(R9*BETA)
      RMENOV = RMENOV+H2
      DO 170 I=1,3
      DO 170 J=1,3
      B(I,J) = BVECT(I)*BVECT(J)
  170 CONTINUE
      DO 180 I=1,3
      DO 180 J=1,3
      D(I,J) = A(I,J)-B(I,J)/RMENOV
  180 CONTINUE
      WRITE(*,11)
      WRITE(*,12)D
   11 FORMAT( ' D = ')
   12 FORMAT(3f15.3)
      STOP
      END
```

The calculated consistent tangential modulus:

```
D =
      99868.323      50710.778           .000
      50710.778      27969.175           .000
           .000           .000        742.379
Stop - Program terminated.
```

In the Table 1 we have introduced the results of the iterative procedure of the test program, which illustrates the effectiveness of the use the tangential material modulus consistent with the integration procedure (the third column). *Fnorma* is the norm of the residual forces and *ITE* is the number of the iteration step. The first two columns in the table indicate these values using the elastic material modulus $\mathbf{D}^e$ and the classical modulus of continuous integration $\mathbf{D}_{cont}$. At the integration procedures it was required that the convergence criterion *Fnorma* should fall below 0.001.

Table 1 Iteration steps at different modules

Elastic material modulus D_e	Modulus D_{cont}	Consistent modulus D_{cons}
ite = 1 Fnorma = 81.9361 ite = 2 Fnorma = 45.0528 ite = 3 Fnorma = 24.1507 ite = 4 Fnorma = 12.7373 ite = 5 Fnorma = 6.64959 ite = 6 Fnorma = 3.44982 ite = 7 Fnorma = 1.78315 ite = 8 Fnorma = .919702 ite = 9 Fnorma = .473786 ite =10 Fnorma = .243908 ite =11 Fnorma = .125520 ite =12 Fnorma = .645826E-01 ite =13 Fnorma = .332255E-01 ite =14 Fnorma = .170925E-01 ite =15 Fnorma = .879277E-02 ITERATION NUMBER = 15 The limit number of iterations was exceeded	ite = 1 Fnorma = 81.9361 ite = 2 Fnorma = 18.1286 ite = 3 Fnorma = 6.25226 ite = 4 Fnorma = 1.70708 ite = 5 Fnorma = .430342 ite = 6 Fnorma = .104781 ite = 7 Fnorma = .250916E-01 ite = 8 Fnorma = .595902E-02 ite = 9 Fnorma = .140928E-02 ite =10 Fnorma = .332571E-03	ite = 1 Fnorma = 81.9361 ite = 2 Fnorma = 4.06090 ite = 3 Fnorma = .142896E-02 ite = 4 Fnorma = .113480E-06

[1] Benča Š.: Aplikovaná nelineárna mechanika kontinua, Vydavateľstvo 1000knih.sk, 2013
(Applied nonlinear continuum mechanics – in Slovak, extracts on *mkp-fem.sk*)